// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/interfaces/IStakingForClaims.sol



// SPDX-License-Identifier: CC0
// Copyright (c) 2022 unReal Accelerator, LLC (https://unrealaccelerator.io)
pragma solidity ^0.8.9;

/// @title IStakingForClaims
/// @author jason@unrealaccelerator.io
/// @notice Interface for getting count of claims

interface IStakingForClaims {
    function balanceOf(address account) external view returns (uint256);
}

// File: contracts/ShillingTokenClaims.sol



// SPDX-License-Identifier: CC0
// Copyright (c) 2022 unReal Accelerator, LLC (https://unrealaccelerator.io)
pragma solidity ^0.8.19;

/**
 * @title ShillingTokenClaims
 * @dev A contract for managing claims of Shilling tokens from staking contracts.
 * @author jason@unrealaccelerator.io
 */

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./interfaces/IStakingForClaims.sol";

contract ShillingTokenClaims is ReentrancyGuard, Ownable {
    IERC20 public tokenContract; // ERC20 token contract for Shilling tokens
    IStakingForClaims public stakingCOKContract; // Staking contract for COK tokens
    IStakingForClaims public stakingHCOKContract; // Staking contract for HCOK tokens

    uint256 public ethPerThreeShillings = 0.22 ether; // Default eth amount for 3 shilling

    // mapping to track claimed amounts per address
    mapping(address => uint256) public claimedAmountForCOK;
    mapping(address => uint256) public claimedAmountForHCOK;

    ///////// Security

    address public administrator; // Address with administrative privileges
    bool public claimAllowed; // Used to control claim events
    bool public purchaseAllowed; // Used to control shilling purchase events

    ///////// Errors

    error NotAuthorized(); // Error thrown when caller is not authorized
    error NoBalanceToWithdraw(); // Error thrown when attempting to withdraw with no balance
    error ClaimNotAllowed(); // Error thrown with attempting to claim but claiming is not allowed
    error InvalidNumberOfTokens();
    error PurchaseNotAllowed();
    error AddressHasNotEnoughClaims();
    error InsufficientTokens();
    error IncorrectAmountOfETH();

    /////////  Events

    event ClaimAllowed(bool indexed state);
    event PurchaseAllowed(bool indexed state);
    event Claimed(address indexed user, uint256 amount);
    event ClaimsPurchased(
        address indexed user,
        uint256 amount,
        uint256 totalEth
    );

    ///////// Modifiers

    /**
     * @dev Modifier to check for authorized roles (Owner or Administrator)
     */
    modifier onlyAuthorized() {
        validateAuthorized();
        _;
    }

    function validateAuthorized() private view {
        if (_msgSender() != owner() && _msgSender() != administrator)
            revert NotAuthorized();
    }

    /**
     * @dev Constructor to initialize the contract with required addresses.
     * @param tokenContract_ Address of the Shilling token contract.
     * @param stakingCOKContract_ Address of the staking contract for COK tokens.
     * @param stakingHCOKContract_ Address of the staking contract for HCOK tokens.
     * @param administrator_ Address with administrative privileges.
     */
    constructor(
        address tokenContract_,
        address stakingCOKContract_,
        address stakingHCOKContract_,
        address administrator_
    ) {
        tokenContract = IERC20(tokenContract_);
        stakingCOKContract = IStakingForClaims(stakingCOKContract_);
        stakingHCOKContract = IStakingForClaims(stakingHCOKContract_);
        administrator = administrator_;
    }

    /**
     * @dev Fallback function to receive Ether.
     * This function is payable and allows the contract to receive ETH.
     */
    receive() external payable {}

    /**
     * @dev Get the current supply of Shilling tokens held by the contract.
     * @return The balance of Shilling tokens held by the contract.
     */
    function shillingSupply() external view returns (uint256) {
        return uint256(tokenContract.balanceOf(address(this)));
    }

    /**
     * @dev Calculates the number of allowed claims based on the staked COK balance.
     * @param account The address of the account to check allowed claims for.
     * @return The number of allowed claims. Each staked COK token grants 1 claim, rounded down to the nearest multiple of 3.
     */
    function allowedClaimsWithCOK(
        address account
    ) public view returns (uint256) {
        uint256 balance = stakingCOKContract.balanceOf(account);
        // 1 claim per staked COK; but only in multiples of 3
        uint256 claims = (balance / 3) * 3;
        return claims;
    }

    /**
     * @dev Calculates the number of remaining claims based on the staked COK balance.
     * @param account The address of the account to check remaining claims for.
     * @return The number of remaining claims. Each staked COK token grants 1 claim, rounded down to the nearest multiple of 3.
     */
    function remainingClaimsWithCOK(
        address account
    ) public view returns (uint256) {
        uint256 balance = allowedClaimsWithCOK(account) -
            claimedAmountForCOK[account];
        return balance;
    }

    /**
     * @dev Get the number of allowed claims for an account using the stakingHCOKContract.
     * @param account The address of the account to check allowed claims for.
     * @return The number of allowed claims based on the staked HCOK balance.
     */
    function allowedClaimsWithHCOK(
        address account
    ) public view returns (uint256) {
        return (stakingHCOKContract.balanceOf(account) * 9);
    }

    /**
     * @dev Calculates the number of remaining claims based on the staked HCOK balance.
     * @param account The address of the account to check remaining claims for.
     * @return The number of allowed remaining based on the staked HCOK balance.
     */
    function remainingClaimsWithHCOK(
        address account
    ) public view returns (uint256) {
        uint256 balance = allowedClaimsWithHCOK(account) -
            claimedAmountForHCOK[account];
        return balance;
    }

    /**
     * @dev Allows an account to claim rewards using COK tokens.
     * The function checks if claiming is allowed, calculates the unclaimed amount based on the account's staked COK balance,
     * transfers the rewards to the account, and emits a `Claimed` event.
     * This function is non-reentrant.
     * @notice Only addresses with unclaimed rewards can execute this function.
     * @notice The contract must have sufficient token balance to cover the claimed rewards.
     * @notice Emits a `Claimed` event upon successful claim.
     */
    function claimWithCOK(uint256 numberOfShillings) external nonReentrant {
        // Get the unclaimed amount of rewards for the caller's account
        uint256 unclaimedAmount = remainingClaimsWithCOK(_msgSender());

        _claim(unclaimedAmount, numberOfShillings);

        claimedAmountForCOK[_msgSender()] =
            claimedAmountForCOK[_msgSender()] +
            numberOfShillings;
    }

    /**
     * @dev Allows an account to claim rewards using HCOK tokens.
     * The function checks if claiming is allowed, calculates the unclaimed amount based on the account's staked HCOK balance,
     * transfers the rewards to the account, and emits a `Claimed` event.
     * This function is non-reentrant.
     * @notice Only addresses with unclaimed rewards can execute this function.
     * @notice The contract must have sufficient token balance to cover the claimed rewards.
     * @notice Emits a `Claimed` event upon successful claim.
     */
    function claimWithHCOK(uint256 numberOfShillings) external nonReentrant {
        // Get the unclaimed amount of rewards for the caller's account
        uint256 unclaimedAmount = remainingClaimsWithHCOK(_msgSender());

        _claim(unclaimedAmount, numberOfShillings);

        // Update the claimed amount for caller's address
        claimedAmountForHCOK[_msgSender()] =
            claimedAmountForHCOK[_msgSender()] +
            numberOfShillings;
    }

    function _claim(uint256 unclaimedAmount, uint256 claimAmount) private {
        // Ensure claiming is allowed
        if (!claimAllowed) revert ClaimNotAllowed();

        // Ensure the account has unclaimed rewards
        if (claimAmount <= 0 || claimAmount > unclaimedAmount)
            revert AddressHasNotEnoughClaims();

        // Calculate the balance of the contract's token holdings
        uint256 balance = uint256(tokenContract.balanceOf(address(this)));

        // Calculate the reward to be transferred to the account in wei
        uint256 rewardWei = claimAmount * 10 ** 18;

        // Ensure the contract has sufficient tokens to cover the reward
        if (rewardWei > balance) revert InsufficientTokens();

        // Transfer the reward tokens to the account
        tokenContract.transfer(_msgSender(), rewardWei);
        // Emit a Claimed event to indicate a successful claim
        emit Claimed(_msgSender(), claimAmount);
    }

    /**
     * @dev Allows an account to purchase rewards using ETH.
     * The function checks if claiming is allowed, calculates the number of claims that can be purchased,
     * verifies that the number of claims is a multiple of 3, calculates the total ETH required,
     * transfers the ETH to the contract, and emits a `ClaimsPurchased` event.
     * This function is non-reentrant.
     * @notice Only addresses can purchase claims in multiples of 3.
     * @notice Emits a `ClaimsPurchased` event upon successful purchase.
     * @param numberOfShillings The number of claims to purchase (must be a multiple of 3).
     */
    function claimWithEth(
        uint256 numberOfShillings
    ) external payable nonReentrant {
        // Ensure claiming is allowed
        if (!purchaseAllowed) revert PurchaseNotAllowed();

        // Verify that the number of claims is a multiple of 3
        if (numberOfShillings % 3 != 0 || numberOfShillings == 0)
            revert InvalidNumberOfTokens();

        // Calculate the total ETH required for the claims
        uint256 totalEthRequired = (numberOfShillings / 3) *
            ethPerThreeShillings;

        // Ensure the sent ETH matches the required amount
        if (msg.value != totalEthRequired) revert IncorrectAmountOfETH();

        // Calculate the balance of the contract's token holdings
        uint256 balance = uint256(tokenContract.balanceOf(address(this)));

        uint256 numberOfShillingsWei = numberOfShillings * 10 ** 18;

        // Ensure the contract has sufficient tokens to cover the reward
        if (int256(numberOfShillingsWei) - int256(balance) > 0)
            revert InsufficientTokens();

        // Transfer the Shillings to the account
        tokenContract.transfer(_msgSender(), uint256(numberOfShillingsWei));

        // Emit a ClaimsPurchased event to indicate a successful purchase
        emit ClaimsPurchased(
            _msgSender(),
            numberOfShillingsWei,
            totalEthRequired
        );
    }

    /**
     * @dev Activate/disable claiming
     * Only authorized addresses can call this function.
     */
    function setClaimAllowed(bool claimAllowed_) external onlyAuthorized {
        claimAllowed = claimAllowed_;
        emit ClaimAllowed(claimAllowed);
    }

    /**
     * @dev Activate/disable purchasing
     * Only authorized addresses can call this function.
     */
    function setPurchaseAllowed(bool purchaseAllowed_) external onlyAuthorized {
        purchaseAllowed = purchaseAllowed_;
        emit PurchaseAllowed(purchaseAllowed);
    }

    /**
     * @dev Set/update the token contract address
     * Only authorized addresses can call this function.
     */
    function setTokenContract(address tokenContract_) external onlyAuthorized {
        tokenContract = IERC20(tokenContract_);
    }

    /**
     * @dev Set/update the COK contract address
     * Only authorized addresses can call this function.
     */
    function setStakingCOKContract(
        address stakingCOKContract_
    ) external onlyAuthorized {
        stakingCOKContract = IStakingForClaims(stakingCOKContract_);
    }

    /**
     * @dev Set/update the HCOK contract address
     * Only authorized addresses can call this function.
     */
    function setStakingHCOKContract(
        address stakingHCOKContract_
    ) external onlyAuthorized {
        stakingHCOKContract = IStakingForClaims(stakingHCOKContract_);
    }

    /**
     * @dev Set/update the eth required to claim 3 shillings
     * Amount must be in wei
     * Only authorized addresses can call this function.
     */
    function setEthPerThreeShillings(
        uint256 ethPerThreeShillings_
    ) external onlyAuthorized {
        ethPerThreeShillings = ethPerThreeShillings_;
    }

    /**
     * @dev Admin function to withdraw unclaimed Shilling tokens after the claiming period.
     * Only authorized addresses can call this function.
     */
    function withdrawRemainingToken() external onlyAuthorized {
        uint256 amount = tokenContract.balanceOf(address(this));
        tokenContract.transfer(owner(), amount);
        delete amount;
    }

    /**
     * @dev Admin function to withdraw ETH from the contract.
     * Only authorized addresses can call this function.
     */
    function withdraw() external onlyAuthorized {
        uint256 balance = address(this).balance;
        if (balance <= 0) revert NoBalanceToWithdraw();

        address payable ownerAddress = payable(owner());
        ownerAddress.transfer(balance);
    }
}